query: #-------------------------------------------------------------------------------
  # Straßen, Plätze und Brücken in Dresden (Übersicht mit Stadtwiki und Themenstadtplan)
  #-------------------------------------------------------------------------------
  
  PREFIX schema: <http://schema.org/>
  #defaultView:Map
  SELECT DISTINCT ?item ?itemLabel ?image 
     (URI(CONCAT("https://stadtplan.dresden.de/?POS-ADR=",?DDstreetID)) as ?DDstreetID_link)   
     (URI(CONCAT("https://www.stadtwikidd.de/wiki/",?stadtwikiDD)) as ?stadtwikiDD_link) 
     ?commons2 ?coordinate ?status ?rgb
     
  WHERE {
    ?item wdt:P8710 ?streetID
    OPTIONAL { ?item wdt:P373 ?commons. }                                  # Commons-Kategorie
    #FILTER(!BOUND(?commons)) #commons leer
    OPTIONAL { ?item wdt:P18 ?image. }                                  # optional: Bild
    OPTIONAL { ?item wdt:P625 ?coordinate. }                          # optional: Koordinate (für Ansicht in Karte)
    SERVICE wikibase:label { bd:serviceParam wikibase:language "de". } # Sprachausgabe deutsch
     
    BIND(BOUND(?commons) && strlen(?commons)>0 as ?check_cat)          # true/false - gibt es eine Commons-Category 
    BIND(BOUND(?image) as ?check_img)                                  # true/false - gibt es ein Bild
     
    BIND(  IF( ?check_img , "image",
           IF( ?check_cat , "category",   
              "no cat")) AS ?status).
  
    BIND(  IF( ?check_img , "7FFF00",
           IF( ?check_cat , "009500",   
              "ff2d00")) AS ?rgb). 
    # https://htmlcolorcodes.com/
    # 7FFF00 neongrün
    # 009500 dunkelgrün
    # FFF000 gelb
    # 9B0058 rot
    # ff2d00 hellrot
    
    OPTIONAL {?item wdt:P8713 ?stadtwikiDD . }
    OPTIONAL {?item wdt:P8710 ?DDstreetID . }
    OPTIONAL {?commons2 	schema:about ?item ;
             schema:isPartOf <https://commons.wikimedia.org/> .
    } 
   
    #?item rdfs:label ?streetname .                                    # Straßennamen an Variable binden
    #FILTER(REGEX(STR(?streetname), "Kon")).                           # Straßennamen filtern
  }